Scripts para gerar Carta de Smith do GRF em formato vetorial

Procedimento:

1. Copiar e colar os três códigos abaixo em três arquivos distintos: xplot.plt, smithplot.plt e smithloop.plt
2. Substituir “sparams_sim.csv” na linha 21 do xplot.plt pelo nome do seu arquivo de dados
3. Rodar o xplot.plt no terminal

==========xplot.plt==========

```
reset;
#set term eps enhanced dashed
set terminal postscript enhanced
set term post eps monochrome

set output "reflexao.eps"

load 'smithplot.plt'

set style line 1 lc rgb 'black' pt 1 ps 1 lt 1 lw 3
set style line 2 lc rgb 'black' pt 1 ps 1 lt 2 lw 3

#set label "800 MHz" at -0.2,-0.5 font "Arial,11"
#set label "1 GHz" at 0.35,0.24 font "Arial,11"
#set label "-30 dBm" at -0.5,0.4 font "Arial,11"
#set label "10 dBm" at 0.5,-0.5 font "Arial,11"
set label "|{/Symbol \107}|=-10 dB" at  -0.1,0.35 #font "BoldArial,14"
set label "F=2.4 GHz" at  -0.08,-0.19

set key
plot "sparams_sim.csv" using 2:3:1 w l lt 1 lc -1 lw 5 title "Simulated", "pontos" using 2:3 w p lc -1 ps 2 lw 3 pt 3 notitle ,"lowpower.s2p" using 2:3 every 6 w l lt 2 lc -1 lw 5 title "Measured"
```

==========smithplot.plt==========

```
reset
unset multiplot
complex(x,y) = x*{1,0}+y*{0,1}
conformal(z) = (z-1)/(z+1)
pi = 4*atan(1)
lblIMRE(x) = 1.05*real(conformal(complex(0,x)))-0.025
lblIMIM(x) = 1.05*imag(conformal(complex(0,x)))
lblRERE(x) = real(conformal(complex(x,0)))+0.005

unset border
unset xtics
unset ytics
unset xlabel
unset ylabel
unset logscale x
```

```
unset logscale y
set nokey
set size square 1,1
set parametric
set xrange[-1.1:1.1]
set yrange[-1.1:1.1]
set samples 1500


set label 1 "0" at lblIMRE(0),lblIMIM(0) font "times,14"
set label 3 "j.2" at lblIMRE(0.12),lblIMIM(0.2) font "times,14"
set label 5 "j.4" at lblIMRE(0.34),lblIMIM(0.4) font "times,14"
set label 7 "j.6" at lblIMRE(0.56),lblIMIM(0.6) font "times,14"
set label 9 "j.8" at lblIMRE(0.8),lblIMIM(0.8) font "times,14"
set label 20 "j1" at lblIMRE(1),lblIMIM(1) font "times,14"
set label 21 "j2" at lblIMRE(2),lblIMIM(2) font "times,14"
set label 23 "j4" at lblIMRE(4),lblIMIM(4) font "times,14"
set label 25 "j6" at lblIMRE(6),lblIMIM(6) font "times,14"

set label 103 "-j.2" at lblIMRE(0.0),lblIMIM(-0.20) font "times,14"
set label 105 "-j.4" at lblIMRE(-0.32),lblIMIM(-0.4) font "times,14"
set label 107 "-j.6" at lblIMRE(-0.54),lblIMIM(-0.6) font "times,14"
set label 109 "-j.8" at lblIMRE(-0.78),lblIMIM(-0.8) font "times,14"
set label 1020 "-j1" at lblIMRE(-1),lblIMIM(-1) font "times,14"
set label 1021 "-j2" at lblIMRE(-2),lblIMIM(-2) font "times,14"
set label 1023 "-j4" at lblIMRE(-4),lblIMIM(-4) font "times,14"
set label 1025 "-j6" at lblIMRE(-6),lblIMIM(-6) font "times,14"

set label 2002 "0.2" at lblRERE(0.2),0.025 font "times,14"
set label 2004 "0.4" at lblRERE(0.4),0.025 font "times,14"
set label 2006 "0.6" at lblRERE(0.6),0.025 font "times,14"
set label 2010 "1" at lblRERE(1),0.025 font "times,14"
set label 2011 "2" at lblRERE(2),0.025 font "times,14"
set label 2013 "4" at lblRERE(4),0.025 font "times,14"

set style line 1 lc rgb '#404040' pt 1 ps 1 lt 1 lw 2 # grid

set label 30 "{/Symbol \245}" at 1.01,0

#a=-1
a=0
set multiplot
load "smithloop.plt"

plot '-' with circles ls 1
0 0 1
eof
plot '-' with lines ls 1
-1 0
```

```
1 0
eof

plot '-' with circles lc rgb '#424242' lt 3 lw 2
0 0 0.316
eof


#unset multiplot
unset parametric

==============================smithloop.plt================================
   a = a+2
   set trange [-60:60]
   plot real(conformal(complex(a,t))),imag(conformal(complex(a,t))) w l ls 1
   plot real(conformal(complex(0.1*a,t))),imag(conformal(complex(0.1*a,t))) w l ls 1

   set trange [0:10]
   plot real(conformal(complex(t,a))),imag(conformal(complex(t,a))) w l ls 1
   plot real(conformal(complex(t,a))),-imag(conformal(complex(t,a))) w l ls 1

   set trange [0:4]
   if (a==2) || (a==4) set trange [0:1]
   plot real(conformal(complex(t,0.1*a))),imag(conformal(complex(t,0.1*a))) w l ls 1
   plot real(conformal(complex(t,0.1*a))),-imag(conformal(complex(t,0.1*a))) w l ls 1

   if(a<10) reread
================================================================================
```